Pensieve Header: The β invariants of braids.

Utilities

```
In[1]:= βSimplify = Factor;
SetAttributes[βCollect, Listable];
βCollect[B[ω_, μ_]] := B[
   βSimplify[ω],
   Collect[μ, _h, Collect[#, _t, βSimplify] &]
  ];
(* "L" for "Labels" *)
hL[β_] := Union[Cases[β, h[s_] :> s, Infinity]];
tL[β_] := Union[Cases[β, t[s_] | c_s_ :> s, Infinity]];
dL[β_] := Union[hL[β], tL[β]];
βForm[B[ω_, μ_]] := Module[
   {tails, heads, mat},
   tails = tL[B[ω, μ]]; heads = hL[B[ω, μ]];
   mat = Outer[βSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
   PrependTo[mat, t /@ tails];
   mat = Prepend[Transpose[mat], Prepend[h /@ heads, ω]];
   MatrixForm[mat]
  ];
βForm[else_] := else /. β_B :> βForm[β];
B /: B[ω1_, β1_] == B[ω2_, β2_] := (ω1 == ω2) && (β1 == β2);
```

The Meta-Cross-Product

The “Tails” meta-group

```
In[10]:= tm[x_, y_, z_][β_] := β /. {t[x] → t[z], t[y] → t[z], c_x → c_z, c_y → c_z};
tΔ[z_, x_, y_][β_] := βCollect[β /. {t[z] → t[x] + t[y], c_z → c_x + c_y}];
tη[x_][β_] := βCollect[(β /. t[x] → 0) /. c_x → 0];
tS[x_][β_] := βCollect[β /. {t[x] → -t[x], c_x → -c_x}];
tA[_][β_] := βCollect[β];
tP[rules___Rule][β_] := βCollect[
   β /. {t[x_] :> t[x /. {rules}], c_x_ :> c_(x /. {rules})}
  ];
```

The “Heads” meta-group

```
In[16]:= hm[x_, y_, z_][B[ω_, μ_]] := Module[
           {γx = D[μ, h[x]], γy = D[μ, h[y]], M = μ /. h[x] | h[y] → 0},
           B[ω, M + h[z] (γx + γy + (γx /. t[i_] ⧴ c_i) γy)] // βCollect
          ];
         hΔ[z_, x_, y_][β_] := βCollect[β /. h[z] → h[x] + h[y]];
         hη[x_][β_] := βCollect[β /. h[x] → 0];
         hS[x_][B[ω_, μ_]] := Module[{γ},
           γ = 1 + D[μ, h[x]] /. t[s_] ⧴ c_s;
           βCollect[B[ω, μ /. h[x] → -h[x] / γ]]
          ];
         hA[x_][β_] := hS[x][β];
         hP[rules___Rule][β_] := βCollect[β /. h[x_] ⧴ h[x /. {rules}]];
```

The TH → HT Swap

```
In[22]:= thswap[x_, y_][B[ω_, μ_]] := Module[
           {α, β, γ, δ, ϵ},
           α = Coefficient[μ, h[x] t[y]];
           β = D[μ, t[y]] /. h[x] → 0;
           γ = D[μ, h[x]] /. t[y] → 0;
           δ = μ /. h[x] | t[y] → 0;
           ϵ = 1 + c_y α;
           B[ω * ϵ, Plus[
              α (1 + (γ /. t[i_] ⧴ c_i) / ϵ) h[x] t[y],
              β (1 + (γ /. t[i_] ⧴ c_i) / ϵ) t[y],
              γ / ϵ h[x],
              δ - c_y / ϵ γ * β
             ]] // βCollect
          ];
```

The “double” meta-group

```
In[23]:= dm[x_, y_, z_][β_] := β // thswap[y, x] // hm[x, y, z] // tm[x, y, z];
```

The “external” product

```
In[24]:= B /: B[ω1_, μ1_] B[ω2_, μ2_] := B[ω1 * ω2, μ1 + μ2];
```

The R-Matrix

```
In[25]:= R[x_, y_] := B[1, (E^c_x - 1) / c_x * t[x] h[y]];
         Rinv[x_, y_] := B[1, (E^(-c_x) - 1) / c_x * t[x] h[y]];
```

The Group A_n

```
In[27]:= ar[i_, j_] := t[i] h[j];
         htswap[y_, x_][β_] := β // hS[x] // thswap[y, x] // hS[x];
         dA[x_][β_] := β // tA[x] // htswap[x, x] // hA[x];
         dS[x_][β_] := β // tS[x] // htswap[x, x] // hS[x];
```

In[31]:=

```
Unprotect[NonCommutativeMultiply];
β_ ** ν_ := Module[
   {ρ, σ, labels},
   ρ = β * (ν /. {h[s_] :→ h[σ[s]], t[s_] :→ t[σ[s]], c_s_ :→ c_σ[s]});
   labels = Union[Cases[{β, ν}, h[s_] | t[s_] | c_s_ :→ s, Infinity]];
   Do[
    ρ = ρ // dm[s, σ[s], s],
    {s, labels}
   ];
   ρ
  ];
B /: Inverse[B[ω_, μ_]] := Module[
   {ρ = B[1, μ]},
   Do[ρ = ρ // dA[s], {s, Union[hL[ρ], tL[ρ]]}];
   ReplacePart[ρ, 1 → 1 / ω] // βCollect
  ];
```

In[44]:=

```
{
   R[1, 2],
   R[1, 2] ** R[2, 1],
   R[1, 2] ** R[2, 3],
   R[2, 3] ** R[1, 2],
   R[1, 2] ** R[2, 3] ** R[1, 2],
   R[1, 2] ** R[2, 3] ** R[1, 2] ** R[2, 1]
  } // βForm // ColumnForm
```

Out[44]=

$$\begin{pmatrix} 1 & h[2] \\ t[1] & \frac{-1+e^{c_1}}{c_1} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[1] & h[2] \\ t[1] & 0 & \frac{e^{c_2}(-1+e^{c_1})}{c_1} \\ t[2] & \frac{-1+e^{c_2}}{c_2} & -\frac{(-1+e^{c_1})(-1+e^{c_2})}{c_2} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & 0 \\ t[2] & 0 & \frac{-1+e^{c_2}}{c_2} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & -\frac{(-1+e^{c_1})(-1+e^{c_2})}{c_1} \\ t[2] & 0 & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{(-1+e^{c_1})(1+e^{c_1})}{c_1} & -\frac{(-1+e^{c_1})(-1+e^{c_2})}{c_1} \\ t[2] & 0 & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}$$

$$\begin{pmatrix} 1 & h[1] & h[2] & h[3] \\ t[1] & 0 & \frac{e^{c_2}(-1+e^{c_1})(1+e^{c_1})}{c_1} & -\frac{e^{c_2}(-1+e^{c_1})(-1+e^{c_2})}{c_1} \\ t[2] & \frac{-1+e^{c_2}}{c_2} & -\frac{(-1+e^{c_1})(1+e^{c_1})(-1+e^{c_2})}{c_2} & \frac{(-1+e^{c_2})(1-e^{c_2}+e^{c_1+c_2})}{c_2} \end{pmatrix}$$

In[42]:= $\frac{-1+e^{c_2}}{c_2} + -\frac{(-1+e^{c_1})\,(-1+e^{c_2})}{c_2}$ // Simplify

Out[42]= $-\frac{(-2+e^{c_1})\,(-1+e^{c_2})}{c_2}$

In[43]:= $e^{c_2}\,(-1+e^{c_1})\,(1+e^{c_1}) - (-1+e^{c_1})\,(1+e^{c_1})\,(-1+e^{c_2})$ // Simplify

Out[43]= $-1+e^{2\,c_1}$